AtCoder Beginners Selectionの実施記録(Visual Basic)

## 概要

表題のプログラミング言語でAtCoder Beginners Selectionを実施記録した際の記録です
基本文法の学習が目的です

> https://atcoder.jp/contests/abs
> AtCoder Beginners Selectionとは？
> このコンテストは、「AtCoderに登録したけど何をしていいか分からない・・・！」という人に向けて作られた、初心者向け問題集です。
> 問題は、dr.kenさんの、「AtCoder に登録したら次にやること ～ これだけ解けば十分闘える！過去問精選 10 問 ～」から選出しています。

基本的には以下ができれば大体の問題を解けるようになっている。

- 標準入力
- 標準出力
- 四則演算
- 条件分岐
- 文字列操作
- ループ
- 配列の操作
- 全通りの探索
- 組み込み関数等の利用
- ソート
- 集合
- 二次元配列の操作

## PracticeA - Welcome to AtCoder 

[PracticeA - Welcome to AtCoder](https://atcoder.jp/contests/abs/tasks/practice_1)

> 問題文
> 高橋君はデータの加工が行いたいです。
> 整数 a,b,cと、文字列 s が与えられます。 a+b+c の計算結果と、文字列 s を並べて表示しなさい。
>
> 制約
1≤a,b,c≤1,000
1≤∣s∣≤100
>
> 入力
> 入力は以下の形式で与えられる。
> a
> b c
> s
>
> 出力
> a+b+c と s を空白区切りで 1 行に出力せよ。

基本的な標準入力、標準出力、四則演算ができればOK

```visualbasic:main.vb
module main
  sub Main(byval args() as string)
    dim a,b,c as integer
    dim s,wk as string
    dim arr() as string
    wk = console.readline()
    a = integer.parse(wk)
    arr = console.readline().split(" ")
    b = integer.parse(arr(0))
    c = integer.parse(arr(1))
    s = console.readline()
    console.writeline((a+b+c).tostring() & " " & s)
  end sub
end module
```

## ABC086A - Product 

[ABC086A - Product](https://atcoder.jp/contests/abs/tasks/abc086_a)

>問題文
シカのAtCoDeerくんは二つの正整数 a,b を見つけました。 a と b の積が偶数か奇数か判定してください。
>
>制約
>1 ≤ a,b ≤ 10000
>a,b は整数
>
>入力
>入力は以下の形式で標準入力から与えられる。
>a b
>
>出力
積が奇数なら Odd と、 偶数なら Even と出力せよ。

基本的な標準入力、標準出力、四則演算、条件分岐ができればOK

```visualbasic:main.vb
module main
  sub Main(byval args() as string)
    ' 入力を受け取る
    dim a,b as integer
    dim arr() as string
    arr = console.readline().split(" ")
    a = integer.parse(arr(0))
    b = integer.parse(arr(1))

    ' aとbの積が偶数か奇数か判定する
    If (a * b) Mod 2 = 0 Then
      Console.WriteLine("Even")
    Else
      Console.WriteLine("Odd")
    End If
  end sub
end module
```

## ABC081A - Placing Marbles 

[ABC081A - Placing Marbles ](https://atcoder.jp/contests/abs/tasks/abc081_a)

> 問題文
>すぬけ君は 1,2,3 の番号がついた 3 つのマスからなるマス目を持っています。各マスには 0 か 1 が書かれており、マス i には si が書かれています。
>すぬけ君は 1 が書かれたマスにビー玉を置きます。 ビー玉が置かれるマスがいくつあるか求めてください。
>
>制約
>s1 ,s2 ,s3 は 1 あるいは 0
>
>入力
>入力は以下の形式で標準入力から与えられる。
>s1s2s3 
>
>出力
>答えを出力せよ。
>
>入力例 1
>101
>
>出力例 1
>2


基本的な標準入力、標準出力、文字列操作ができればOK

```visualbasic:main.vb
Module MainModule

    Sub Main()
        Dim s As String = Console.ReadLine()
        Dim count As Integer

        For Each c In s
          If c = "1" Then
             count = count + 1
          End If
        Next
        Console.WriteLine(count)
    End Sub
End Module
```

## ABC081B - Shift only 
[ABC081B - Shift only](https://atcoder.jp/contests/abs/tasks/abc081_b)

> 問題文
>黒板に N 個の正の整数 A1,...,AN が書かれています．
>すぬけ君は，黒板に書かれている整数がすべて偶数であるとき，次の操作を行うことができます．
>黒板に書かれている整数すべてを，2 で割ったものに置き換える．
>すぬけ君は最大で何回操作を行うことができるかを求めてください．
>
>制約
>1≤N≤200
>1≤Ai≤10^9
>
>入力
>入力は以下の形式で標準入力から与えられる。
>N
>A1 A2 ... AN
>
>出力
>すぬけ君は最大で何回操作を行うことができるかを出力せよ．
>
>入力例 1 
>3
>8 12 40
>
>出力例 1 
>2

基本的な標準入力、標準出力、四則演算、ループ、条件分岐、配列の操作ができればOK

```visualbasic:main.vb
Module Main
  Sub Main()
    Dim n As Integer = Console.ReadLine() ' N を入力
    Dim a As Integer() = Console.ReadLine().Split(" ").Select(Function(x) Integer.Parse(x)).ToArray() ' A1, A2, ..., AN を入力
    Dim cntmin As Integer = 10000
    Dim cnt As Integer = 0 ' 操作回数をカウントする変数
    For i As Integer = 0 To n - 1 ' A1, A2, ..., AN の各整数について
      cnt = 0
      While a(i) Mod 2 = 0 ' a(i) が偶数のとき
        a(i) /= 2 ' a(i) を 2 で割る
        cnt += 1 ' 操作回数をカウント
      End While
      if cnt < cntmin Then
         cntmin = cnt
      End If
    Next
    Console.WriteLine(cntmin) ' 操作回数を出力
  End Sub
End Module
```

## ABC087B - Coins
[ABC087B - Coins](https://atcoder.jp/contests/abs/tasks/abc087_b)

> 問題文
>あなたは、500 円玉を A 枚、100 円玉を B 枚、50 円玉を C 枚持っています。 これらの硬貨の中から何枚かを選び、合計金額をちょうど X 円にする方法は何通りありますか。
>同じ種類の硬貨どうしは区別できません。2 通りの硬貨の選び方は、ある種類の硬貨についてその硬貨を選ぶ枚数が異なるとき区別されます。
>
>制約
>0≤A,B,C≤50
>A+B+C≥1
>50≤X≤20,000
>A,B,C は整数である
>X は 50 の倍数である
>
>入力
>入力は以下の形式で標準入力から与えられる。
>A
>B
>C
>X
>
>出力
>硬貨を選ぶ方法の個数を出力せよ。

基本的な標準入力、標準出力、四則演算、ループ、条件分岐、配列の操作、全通りの探索ができればOK

```visualbasic:main.vb
Module Main
  Sub Main()
    ' 入力を受け取る
    Dim A As Integer = Console.ReadLine()
    Dim B As Integer = Console.ReadLine()
    Dim C As Integer = Console.ReadLine()
    Dim X As Integer = Console.ReadLine()
    
    ' 硬貨を選ぶ方法の個数をカウントする変数
    Dim count As Integer = 0
    
    ' 硬貨を選ぶ全てのパターンを試す
    For i = 0 To A
      For j = 0 To B
        For k = 0 To C
          ' 選んだ硬貨の合計金額がX円と等しければ、カウントを1増やす
          If 500*i + 100*j + 50*k = X Then
            count += 1
          End If
        Next
      Next
    Next
    
    ' 結果を出力する
    Console.WriteLine(count)
  End Sub
End Module
```

## ABC083B - Some Sums
[ABC083B - Some Sums](https://atcoder.jp/contests/abs/tasks/abc083_b)

>問題文
>1 以上 N 以下の整数のうち、10 進法での各桁の和が A 以上 B 以下であるものの総和を求めてください。
>
>制約
>1≤N≤10^4 
>1≤A≤B≤36
>入力はすべて整数である
>
>入力
>入力は以下の形式で標準入力から与えられる。
>N A B
>
>出力
>1 以上 N 以下の整数のうち、10 進法での各桁の和が A 以上 B 以下であるものの総和を出力せよ。

基本的な標準入力、標準出力、四則演算、ループ、条件分岐、配列の操作、組み込み関数の利用、ができればOK

```visualbasic:main.vb
Module Main
  Sub Main()
    Dim n, a, b As Integer

    ' 入力を受け取る
    ' 入力を受け取る
    Dim c As Integer() = Console.ReadLine().Split(" ").Select(Function(x) Integer.Parse(x)).ToArray()
    n = c(0)
    a = c(1)
    b = c(2)

    ' 総和を求める
    Dim sum As Integer = 0
    For i = 1 To n
        ' 各桁の和を求める
        Dim digitsSum As Integer = 0
        Dim j = i
        While j > 0
            digitsSum += j Mod 10
            j \= 10
        End While

        ' 各桁の和がA以上B以下である場合は、
        ' 総和に加算する
        If a <= digitsSum AndAlso digitsSum <= b Then
            sum += i
        End If
    Next

    ' 結果を出力する
    Console.WriteLine(sum)
  End Sub
End Module
```

## ABC088B - Card Game for Two
[ABC088B - Card Game for Two](https://atcoder.jp/contests/abs/tasks/abc088_b)

> 問題文
>N 枚のカードがあります. i 枚目のカードには, ai という数が書かれています.
>Alice と Bob は, これらのカードを使ってゲームを行います. ゲームでは, Alice と Bob が交互に 1 枚ずつカードを取っていきます. Alice が先にカードを取ります.
>2 人がすべてのカードを取ったときゲームは終了し, 取ったカードの数の合計がその人の得点になります. 2 人とも自分の得点を最大化するように最適な戦略を取った時, Alice は Bob より何点多く取るか求めてください.
>
>制約
>N は 1 以上 100 以下の整数
>ai(1≤i≤N) は 1 以上 100 以下の整数
>
>入力
>入力は以下の形式で標準入力から与えられる.
>N
>a1 a2 a3 ... aN
>
>出力
>両者が最適な戦略を取った時, Alice は Bob より何点多く取るかを出力してください.

基本的な標準入力、標準出力、四則演算、条件分岐、配列の操作、ループ、ソート、ができればOK

```visualbasic:main.vb
Module Main
  Sub Main() ' 入力を受け取る
    Dim n As Integer = Console.ReadLine()
    Dim cards As Integer() = Console.ReadLine().Split(" "c).Select(Function(x)    Integer.Parse(x)).ToArray()

    ' Alice と Bob の合計得点
    Dim aliceTotal As Integer = 0
    Dim bobTotal As Integer = 0

    ' 各カードを降順にソートする
    Array.Sort(cards, Function(x, y) y.CompareTo(x))

    ' 先手が Alice であるため、奇数番目のカードは Alice、偶数番目のカードは Bob が取る
    For i As Integer = 0 To n - 1
        If i Mod 2 = 0 Then
            aliceTotal += cards(i)
        Else
            bobTotal += cards(i)
        End If
    Next

    ' Alice が Bob より多く取ったかどうかを出力する
    Console.WriteLine(aliceTotal - bobTotal)
  End Sub
End Module
```

## ABC085B - Kagami Mochi 
[ABC085B - Kagami Mochi](https://atcoder.jp/contests/abs/tasks/abc085_b)

> 問題文
>X 段重ねの鏡餅 (X≥1) とは、X 枚の円形の餅を縦に積み重ねたものであって、どの餅もその真下の餅より直径が小さい（一番下の餅を除く）もののことです。例えば、直径 10、8、6 センチメートルの餅をこの順に下から積み重ねると 3 段重ねの鏡餅になり、餅を一枚だけ置くと 1 段重ねの鏡餅になります。
>ダックスフンドのルンルンは N 枚の円形の餅を持っていて、そのうち i 枚目の餅の直径は diセンチメートルです。これらの餅のうち一部または全部を使って鏡餅を作るとき、最大で何段重ねの鏡餅を作ることができるでしょうか。
>
>制約
>1≤N≤100
>1≤di≤100
>入力値はすべて整数である。
>
>入力
>入力は以下の形式で標準入力から与えられる。
>N
>d1
>​:
>dN
>
>​出力
>作ることのできる鏡餅の最大の段数を出力せよ。
>
>入力例 1 
>4
>10
>8
>8
>6
>
>出力例 1 
>3


基本的な標準入力、標準出力、四則演算、ループ、配列、集合、ができればOK

```visualbasic:main.vb
Module Main
    Sub Main()
        ' N 枚の餅を持っているとする
        Dim N As Integer = Console.ReadLine()

        ' 餅の直径を配列に格納する
        Dim d(N - 1) As Integer
        For i As Integer = 0 To N - 1
            d(i) = Integer.Parse(Console.ReadLine())
        Next

        ' 餅の直径を小さい順に並び替える
        Array.Sort(d)

        ' 鏡餅を作るために使用する餅の数
        Dim count As Integer = 1
        Dim d_max As Integer = d(0)

        ' 鏡餅を作るために使用する餅の数をカウントする
        For i As Integer = 0 To N - 1
            ' 直径が小さい順に餅を選択する
            If d(i) > d_max Then
                count += 1
                d_max = d(i)
            End If
        Next

        ' 作ることができる鏡餅の最大の段数を出力する
        Console.WriteLine(count)
    End Sub
End Module
```

## ABC085C - Otoshidama
[ABC085C - Otoshidama](https://atcoder.jp/contests/abs/tasks/abc085_c)

>問題文
>日本でよく使われる紙幣は、10000 円札、5000 円札、1000 円札です。以下、「お札」とはこれらのみを指します。
>青橋くんが言うには、彼が祖父から受け取ったお年玉袋にはお札が N 枚入っていて、合計で Y 円だったそうですが、嘘かもしれません。このような状況がありうるか判定し、ありうる場合はお年玉袋の中身の候補を一つ見つけてください。なお、彼の祖父は十分裕福であり、お年玉袋は十分大きかったものとします。
>
>制約
>1≤N≤2000
>1000≤Y≤2×10^7
>N は整数である。
>Y は 1000 の倍数である。
>
>入力
>入力は以下の形式で標準入力から与えられる。
>N Y
>
>出力
>N 枚のお札の合計金額が Y 円となることがありえない場合は、-1 -1 -1 と出力せよ。
>N 枚のお札の合計金額が Y 円となることがありうる場合は、そのような N 枚のお札の組み合わせの一例を「10000 円札 x 枚、5000 円札 y 枚、1000 円札 z 枚」として、x、y、z を空白で区切って出力せよ。複数の可能性が考えられるときは、そのうちどれを出力してもよい。
>
>入力例 1 
>9 45000
>
>出力例 1 
>4 0 5

基本的な標準入力、標準出力、四則演算、ループができればOK

```visualbasic:main.vb
Module Main
  Sub Main()
    Dim N, YY As Integer
    Dim c As Integer() = Console.ReadLine().Split(" ").Select(Function(x) Integer.Parse(x)).ToArray()
    N = c(0)
    YY= c(1)

    Dim ans As String = "-1 -1 -1"
    For x As Integer = 0 To N
      For y As Integer = 0 To N - x
        Dim z As Integer = N - x - y
        If 10000 * x + 5000 * y + 1000 * z = YY Then
          ans = $"{x} {y} {z}"
          Exit For
        End If
      Next
    Next

    Console.WriteLine(ans)
  End Sub
End Module
```

## ABC049C - 白昼夢
[ABC049C - 白昼夢](https://atcoder.jp/contests/abs/tasks/arc065_a)

>問題文
>英小文字からなる文字列 S が与えられます。 Tが空文字列である状態から始め、以下の操作を好きな回数繰り返すことで S=T とすることができるか判定してください。
>
>制約
>1≦∣S∣≦10^5
>S は英小文字からなる。
>
>入力
>入力は以下の形式で標準入力から与えられる。
>S
>
>出力
>S=T とすることができる場合 YES を、そうでない場合 NO を出力せよ。
>
>入力例 1 
>erasedream
>
>出力例 1 
>YES

基本的な標準入力、標準出力、文字列操作ができればOK

```visualbasic:main.vb
Module Main
  Sub Main()
    Dim S As String = Console.ReadLine()

    ' T の末尾に dream dreamer erase eraser のいずれかを追加する操作を
    ' 繰り返し行い、S が T に等しくなるまで繰り返す。
    While S.Length > 0
        ' T の末尾から dreamer を取り除く
        If S.EndsWith("dreamer") Then
            S = S.Substring(0, S.Length - 7)
        ' T の末尾から eraser を取り除く
        ElseIf S.EndsWith("eraser") Then
            S = S.Substring(0, S.Length - 6)
        ' T の末尾から dream を取り除く
        ElseIf S.EndsWith("dream") Then
            S = S.Substring(0, S.Length - 5)
        ' T の末尾から erase を取り除く
        ElseIf S.EndsWith("erase") Then
            S = S.Substring(0, S.Length - 5)
        Else
            ' T の末尾に dream dreamer erase eraser のいずれも追加できない場合は、
            ' S=T とすることができないので NO を出力する。
            Console.WriteLine("NO")
            Return
        End If
    End While

    ' S=T とすることができるので YES を出力する。
    Console.WriteLine("YES")
  End Sub
End Module
```

## ABC086C - Traveling
[ABC086C - Traveling](https://atcoder.jp/contests/abs/tasks/arc089_a)

> 問題文
>シカのAtCoDeerくんは二次元平面上で旅行をしようとしています。 AtCoDeerくんの旅行プランでは、時刻 0 に 点 (0,0) を出発し、 1 以上 N 以下の各 i に対し、時刻tiに 点 (xi,yi) を訪れる予定です。
>AtCoDeerくんが時刻 t に 点 (x,y) にいる時、 時刻 t+1 には 点 (x+1,y), (x−1,y), (x,y+1), (x,y−1) のうちいずれかに存在することができます。 その場にとどまることは出来ないことに注意してください。 AtCoDeerくんの旅行プランが実行可能かどうか判定してください。
>
>制約
>1 ≤ N ≤ 10^5 
>0 ≤ xi ≤ 10^5
>0 ≤ yi ≤ 10^5
>1 ≤ ti ≤ 10^5
>ti < ti+1 (1 ≤ i ≤ N−1)
>入力は全て整数
>
>入力
>入力は以下の形式で標準入力から与えられる。
>N
>t1 x1 y1
>​t2 x2 y2 
>:
>tN xN yN
>
>出力
>旅行プランが実行可能ならYesを、不可能ならNoを出力してください。
>
>入力例 1 
>2
>3 1 2
>6 1 1
>
>出力例 1 
>Yes
>
>入力例 2 
>1
>2 100 100
>
>出力例 2 
>No
>(0,0) にいる状態から 2 秒後に (100,100) にいるのは不可能です。
>
>入力例 3 
>2
>5 1 1
>100 1 1
>
>出力例 3 
>No



基本的な標準入力、標準出力、四則演算、ループ、二次元配列の操作、ができればOK

```visualbasic:main.vb
Module Main
    Sub Main()
        Dim n As Integer = Console.ReadLine()
        Dim p As Point = New Point(0, 0, 0)
        Dim canTravel As Boolean = True
        For i As Integer = 1 To n
            Dim line As String() = Console.ReadLine().Split(" "c)
            Dim t As Integer = Integer.Parse(line(0))
            Dim x As Integer = Integer.Parse(line(1))
            Dim y As Integer = Integer.Parse(line(2))
            Dim d As Integer = Math.Abs(x - p.x) + Math.Abs(y - p.y)
            If d > t - p.t Or (t - p.t - d) Mod 2 = 1 Then
                canTravel = False
                Exit For
            End If
            p = New Point(t, x, y)
        Next
        Console.WriteLine(If(canTravel, "Yes", "No"))
    End Sub

    Class Point
        Public t As Integer
        Public x As Integer
        Public y As Integer
        Public Sub New(t As Integer, x As Integer, y As Integer)
            Me.t = t
            Me.x = x
            Me.y = y
        End Sub
    End Class
End Module
```

# 参考

https://atcoder.jp/contests/abs